# NAG Library Routine Document

# C05ADF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

C05ADF locates a zero of a continuous function in a given interval by a combination of the methods of nonlinear interpolation, linear extrapolation and bisection.

## 2 Specification

```
SUBROUTINE C05ADF (A, B, EPS, ETA, F, X, IFAIL)

INTEGER            IFAIL
REAL (KIND=nag_wp) A, B, EPS, ETA, F, X
EXTERNAL           F
```

## 3 Description

C05ADF attempts to obtain an approximation to a simple zero of the function $f(x)$ given an initial interval $[a, b]$ such that $f(a) \times f(b) \leq 0$. The same core algorithm is used by C05AZF whose specification should be consulted for details of the method used.

The approximation $x$ to the zero $\alpha$ is determined so that at least one of the following criteria is satisfied:

(i) $|x - \alpha| \leq$ EPS,

(ii) $|f(x)| \leq$ ETA.

## 4 References

Brent R P (1973) *Algorithms for Minimization Without Derivatives* Prentice–Hall

## 5 Parameters

1: A – REAL (KIND=nag_wp) *Input*

*On entry*: $a$, the lower bound of the interval.

2: B – REAL (KIND=nag_wp) *Input*

*On entry*: $b$, the upper bound of the interval.

*Constraint*: B $\neq$ A.

3: EPS – REAL (KIND=nag_wp) *Input*

*On entry*: the termination tolerance on $x$ (see Section 3).

*Constraint*: EPS $>$ 0.0.

4: ETA – REAL (KIND=nag_wp) *Input*

*On entry*: a value such that if $|f(x)| \leq$ ETA, $x$ is accepted as the zero. ETA may be specified as 0.0 (see Section 7).

5: F – REAL (KIND=nag_wp) FUNCTION, supplied by the user. *External Procedure*

F must evaluate the function $f$ whose zero is to be determined.

The specification of F is:

```
FUNCTION F (XX)
REAL (KIND=nag_wp) F
REAL (KIND=nag_wp) XX
```

1: XX – REAL (KIND=nag_wp) *Input*

*On entry*: the point at which the function must be evaluated.

F must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)program from which C05ADF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: X – REAL (KIND=nag_wp) *Output*

*On exit*: the approximation to the zero.

7: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to $0$, $-1$ or $1$. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or $1$ is recommended. If the output of error messages is undesirable, then the value $1$ is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is $0$. **When the value $-1$ or $1$ is used it is essential to test the value of IFAIL on exit.**

*On exit*: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, EPS $\leq 0.0$,
or A $=$ B,
or F(A) $\times$ F(B) $> 0.0$.

IFAIL $= 2$

Too much accuracy has been requested in the computation; that is, the zero has been located to relative accuracy at least $\epsilon$, where $\epsilon$ is the ***machine precision***, but the exit conditions described in Section 3 are not satisfied. It is unsafe for C05ADF to continue beyond this point, but the final value of X returned is an accurate approximation to the zero.

IFAIL $= 3$

A change in sign of $f(x)$ has been determined as occurring near the point defined by the final value of X. However, there is some evidence that this sign-change corresponds to a pole of $f(x)$.

## 7 Accuracy

The levels of accuracy depend on the values of EPS and ETA. If full machine accuracy is required, they may be set very small, resulting in an exit with IFAIL $= 2$, although this may involve many more iterations than a lesser accuracy. You are recommended to set ETA $= 0.0$ and to use EPS to control the accuracy, unless you have considerable knowledge of the size of $f(x)$ for values of $x$ near the zero.

## 8 Further Comments

The time taken by C05ADF depends primarily on the time spent evaluating F (see Section 5).

If it is important to determine an interval of relative length less than $2 \times$ EPS containing the zero, or if F is expensive to evaluate and the number of calls to F is to be restricted, then use of C05AZF is recommended. Use of C05AZF is also recommended when the structure of the problem to be solved does not permit a simple F to be written: the reverse communication facilities of C05AZF are more flexible than the direct communication of F required by C05ADF.

## 9 Example

This example calculates an approximation to the zero of $e^{-x} - x$ within the interval $[0, 1]$ using a tolerance of EPS $=$ 1.0E$-$5.

### 9.1 Program Text

```
!   C05ADF Example Program Text
!   Mark 24 Release. NAG Copyright 2012.

    Module c05adfe_mod

!     C05ADF Example Program Module:
!            Parameters and User-defined Routines

!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                  :: nout = 6
    Contains
      Function f(x)

!       .. Function Return Value ..
        Real (Kind=nag_wp)                  :: f
!       .. Scalar Arguments ..
        Real (Kind=nag_wp), Intent (In)     :: x
!       .. Intrinsic Procedures ..
        Intrinsic                           :: exp
!       .. Executable Statements ..
        f = exp(-x) - x

        Return

      End Function f
    End Module c05adfe_mod
    Program c05adfe

!     C05ADF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: c05adf, nag_wp
      Use c05adfe_mod, Only: f, nout
!     .. Implicit None Statement ..
      Implicit None
!     .. Local Scalars ..
      Real (Kind=nag_wp)                  :: a, b, eps, eta, x
      Integer                             :: ifail
```

```
!     .. Executable Statements ..
      Write (nout,*) 'C05ADF Example Program Results'

      a = 0.0E0_nag_wp
      b = 1.0E0_nag_wp
      eps = 1.0E-5_nag_wp
      eta = 0.0E0_nag_wp

      ifail = -1
      Call c05adf(a,b,eps,eta,f,x,ifail)

      Write (nout,*)

      Select Case (ifail)
      Case (0)
        Write (nout,99999) 'Zero at X =', x
      Case (2,3)
        Write (nout,99999) 'Final point = ', x
      End Select

99999 Format (1X,A,F12.5)
    End Program c05adfe
```

## 9.2 Program Data

None.

## 9.3 Program Results

```
 C05ADF Example Program Results

 Zero at X =     0.56714
```